JFP **13** (5): 935–944, September 2003.
DOI: 10.1017/S0956796802004367 Printed in the United Kingdom

FUNCTIONAL PEARLS

Formatting: a class act

RALF HINZE

Institute of Information and Computing Sciences, Utrecht University,
P.O.Box 80.089, 3508 TB Utrecht, The Netherlands
(*e-mail:* ralf@cs.uu.nl)

1 Introduction

When I was a student, Simula was one of the languages taught in introductory programming language courses and I vividly remember a sticker one of our instructors had attached to the door of his office, saying "Simula does it with class". I guess the same holds for Haskell except that Haskell replaces classes by type classes.

Armed with singleton types, multiple-parameter type classes, and functional dependencies we reconsider a problem raised and solved by Danvy (1998) in a previous pearl. The challenge is to implement a variant of C's *printf* function, called *format* below, in a statically typed language. Here is an interactive session that illustrates the problem:

```
Main⟩ :type format (lit "hello world")
String
Main⟩ format (lit "hello world")
"hello world"
Main⟩ :type format int
Int → String
Main⟩ format int 5
"5"
Main⟩ :type format (int ^ lit " is " ^ str)
Int → String → String
Main⟩ format (int ^ lit " is " ^ str) 5 "five"
"5 is five".
```

The format directive *lit s* means emit *s* literally. The directives *int* and *str* instruct *format* to take an additional argument of the types *Int* and *String*, respectively, which is then shown. The circumflex "^" is used to concatenate two directives.

The type of *format* depends upon its first argument, the format directive. In a language with dependent types, such as Cayenne (Augustsson, 1999), *format* is straightforward to implement. This pearl shows that *format* is equally straightforward to realize in a language like Haskell that allows the definition of values that depend on types. Our solution enjoys nice algebraic properties, and is more direct than Danvy's one (the relation between the two approaches is detailed in section 5).

2 Preliminaries: functors

This section briefly reviews the categorical concept of a functor, which is at the heart of the Haskell implementation of *format*. For our purposes, it is sufficient to think of a *functor* as a combination of a type constructor F of kind $\star \to \star$ and a so-called *mapping function* that lifts a given function of type $A \to B$ to a function of type $F\ A \to F\ B$. In Haskell, the concept of a functor is captured by the following class definition:[1]

$$\begin{array}{l}\textbf{class}\ \mathit{Functor}\ F\ \textbf{where}\\ \quad \mathit{map}\ ::\ (A \to B) \to (F\ A \to F\ B).\end{array}$$

Instances of this class are supposed to satisfy the two *functor laws*:

$$\begin{array}{lcl}\mathit{map}\ \mathit{id} & = & \mathit{id}\\ \mathit{map}\ (\phi \cdot \psi) & = & \mathit{map}\ \phi \cdot \mathit{map}\ \psi.\end{array}$$

Typical examples of functors are container types such as lists or trees. In these cases, the mapping function simply applies its first argument to each element of a given container, leaving its structure intact. However, the notion of functor is by no means restricted to container types. For instance, the functional type $(A \to)$ for fixed A is a functor with the mapping function given by *post* − *composition*.[2]

$$\begin{array}{l}\textbf{instance}\ \mathit{Functor}\ (A \to)\ \textbf{where}\\ \quad \mathit{map}\ \phi\ x\ =\ \phi \cdot x\end{array}$$

For this instance, the functor laws reduce to $\mathit{id} \cdot x = x$ and $(\phi \cdot \psi) \cdot x = \phi \cdot (\psi \cdot x)$. The functor $(A \to)$ will play a prominent rôle in the following sections. In addition, we require the *identity functor* and *functor composition*.

$$\begin{array}{l}\textbf{type}\ \mathit{Id}\ A\ =\ A\\ \textbf{instance}\ \mathit{Functor}\ \mathit{Id}\ \textbf{where}\\ \quad \mathit{map}\ =\ \mathit{id}\\ \textbf{type}\ (F \cdot G)\ A\ =\ F\ (G\ A)\\ \textbf{instance}\ (\mathit{Functor}\ F, \mathit{Functor}\ G) \Rightarrow \mathit{Functor}\ (F \cdot G)\ \textbf{where}\\ \quad \mathit{map}\ =\ \mathit{map} \cdot \mathit{map}\end{array}$$

Again, it is easy to see that the functor laws are satisfied. Furthermore, functor composition is associative and has the identity functor as a unit. As an aside, note that these instance declarations are not legal Haskell since *Id* and "$\cdot$" are not data types defined by **data** or by **newtype**. A data type, however, introduces an additional data constructor which affects the readability of the code. Instead we employ **type** declarations as if they worked as **newtype** declarations. Section 6 describes the necessary amendments to make the code run under GHC or Hugs.

[1] We slightly deviate from Haskell's lexical syntax: both type constructors and type variables are written with an initial upper-case letter (a type variable typically consists of a single upper-case letter) and both value constructors and value variables are written with an initial lower-case letter. This convention helps us to keep values and types apart.

[2] The so-called *operator section* $(A \to)$ denotes the partial application of the infix operator '$\to$' to A.

https://doi.org/10.1017/S0956796802004367 Published online by Cambridge University Press

3 Functional unparsing

The Haskell solution is developed in two steps. In this section we show how to define *format* as a *type-indexed value*. The following section then explains how to implement the type-indexed value using multiple-parameter type classes with functional dependencies.

Recall that the type of *format* depends upon its first argument, the format directive. Clearly, we cannot define such a dependently typed function in Haskell if we represent directives by elements of a single data type, say,

$$\textbf{data}\ \mathit{Dir} = \mathit{lit}\ \mathit{String} \mid \mathit{int} \mid \mathit{str} \mid \mathit{Dir}\ \hat{}\ \mathit{Dir}.$$

However, using Haskell's type classes we can define values that depend on types. To use this feature we must arrange that each directive possesses a distinct type. To this end we introduce the following *singleton types*:

$$\begin{array}{lcl}\textbf{data}\ \mathit{LIT} & = & \mathit{lit}\ \mathit{String}\\ \textbf{data}\ \mathit{INT} & = & \mathit{int}\\ \textbf{data}\ \mathit{STR} & = & \mathit{str}\\ \textbf{data}\ D_1\ \hat{}\ D_2 & = & D_1\ \hat{}\ D_2.\end{array}$$

Strictly speaking, *LIT* is not a singleton type since it accommodates more than one element. This is unproblematic, however, since the type of *format* does not depend on the argument of *lit*. Given these declarations, the directive $\mathit{int}\ \hat{}\ \mathit{lit}\ \texttt{" is "}\ \hat{}\ \mathit{str}$, for instance, has type $\mathit{INT}\ \hat{}\ \mathit{LIT}\ \hat{}\ \mathit{STR}$: the structure of the directive is mirrored at the type level. As an aside, note that the type constructor "$\hat{}$", which takes singleton types to singleton types, is isomorphic to the type of pairs. We could have used pairs in the first place but the right-associative infix data constructor "$\hat{}$" saves some parentheses.

We can now define *format* as a type-indexed value of type

$$\mathit{format}_D\ ::\ D \to \mathit{Format}_D\ \mathit{String}.$$

That is, format_D takes a directive of type D and returns "something" of *String* where 'something' is determined by D in the following way:

$$\begin{array}{lcl}\mathit{Format}_{D::\star} & :: & \star \to \star\\ \mathit{Format}_{\mathit{LIT}}\ S & = & S\\ \mathit{Format}_{\mathit{INT}}\ S & = & \mathit{Int} \to S\\ \mathit{Format}_{\mathit{STR}}\ S & = & \mathit{String} \to S\\ \mathit{Format}_{D_1\hat{}D_2}\ S & = & \mathit{Format}_{D_1}\ (\mathit{Format}_{D_2}\ S).\end{array}$$

The type Format_D is a so-called *type-indexed type*, a type that depends on a type. It specifies for each of the directives the additional argument(s) *format* has to take. The most interesting clause is probably the last one: the arguments to be added for $D_1\ \hat{}\ D_2$ are the arguments to be added for D_1 followed by the arguments to be added for D_2. The crucial property of Format_D is that it constitutes a functor. This

https://doi.org/10.1017/S0956796802004367 Published online by Cambridge University Press

can be seen more clearly if we rewrite $Format_D$ in a point-free style.

$$\begin{aligned} Format_{LIT} &= Id \\ Format_{INT} &= (Int \rightarrow) \\ Format_{STR} &= (String \rightarrow) \\ Format_{D_1 \hat{} D_2} &= Format_{D_1} \cdot Format_{D_2} \end{aligned}$$

The implementation of *format* is straightforward except perhaps for the last case.

$$\begin{aligned} &format_D && :: D \rightarrow Format_D\ String \\ &format_{LIT}\ (lit\ s) && = s \\ &format_{INT}\ int && = \lambda i \rightarrow show\ i \\ &format_{STR}\ str && = \lambda s \rightarrow s \\ &format_{D_1 \hat{} D_2}\ (d_1 \hat{}\ d_2) && = format_{D_1}\ d_1 \diamond format_{D_2}\ d_2 \end{aligned}$$

So $format_{INT}\ int$ is just the *show* function and $format_{STR}\ str$ is just the identity on *String*. It remains to define the operator '$\diamond$', which takes an $F\ String$ and a $G\ String$ to a $(F \cdot G)\ String$. We know that $F = Format_{D_1}$ and $G = Format_{D_2}$ but this does not get us any further. The only assumption we may safely take is that F and G are functorial. Fortunately, using the mapping function on F we can turn a value of type $F\ String$ into a value of type $F\ (G\ String)$ provided we supply a function that takes a *String*, say, s to a value of type $G\ String$. We can define a function of the de-; sired type using the mapping function on G provided we supply a function that takes a string, say, t to some resulting string. Now, since we have to concatenate the "output" produced by the two arguments of "$\diamond$", the resulting string must be $s +\!\!+ t$.

$$\begin{aligned} &(\diamond) && :: (Functor\ F, Functor\ G) \Rightarrow F\ String \rightarrow G\ String \rightarrow (F \cdot G)\ String \\ &f \diamond g && = map_F\ (\lambda s \rightarrow map_G\ (\lambda t \rightarrow s +\!\!+ t)\ g)\ f \end{aligned}$$

The operator '$\diamond$' enjoys nice algebraic properties: it is associative and has the empty string, $"" :: Id\ String$, as a unit. The proof of these properties makes use of the functor laws and the fact that $(String, +\!\!+, "")$ forms a monoid. That said it becomes clear that the construction can be readily generalized to arbitrary monoids. As an example, for reasons of efficiency one might want to replace $(String, +\!\!+, "")$ by $(ShowS, \cdot, id)$, which features constant-time concatenation.

4 Functional unparsing in Haskell

How can we implement the type-indexed value $format_D :: D \rightarrow Format_D\ String$ using Haskell's type classes? Clearly, a singleton parameter class won't do since both D and $Format_D$ vary. We are forced to introduce a two argument class that additionally abstracts away from $Format_D$ assigning *format* the general type $D \rightarrow F\ String$. This type is, however, too general since now D and F may vary independently of each other. This additional 'flexibility' is, in fact, not very welcome since it gives rise to severe problems of ambiguity. Fortunately, *functional dependencies* (Jones, 2000) save the day as they allow us to capture the fact that F is determined by D.

$$\begin{aligned} &\textbf{class}\ (Functor\ F) \Rightarrow Format\ D\ F \mid D \rightarrow F\ \textbf{where} \\ &\quad format\ :: D \rightarrow F\ String \end{aligned}$$

https://doi.org/10.1017/S0956796802004367 Published online by Cambridge University Press

The functional dependency $D \to F$ (beware, this is not the function space arrow) constrains the relation to be functional: if there are instances $Format\ D_1\ F_1$ and $Format\ D_2\ F_2$, then $D_1 = D_2$ implies $F_1 = F_2$. Note that F is additionally restricted to be an instance of *Functor*. It remains to supply for each directive D an instance declaration of the form **instance** $Format\ D\ (Format_D)$ **where** $format = format_D$.

```
instance Format LIT Id where
    format (lit s)    = s
instance Format INT (Int →) where
    format int        = λi → show i
instance Format STR (String →) where
    format str        = λs → s
instance (Format D1 F1, Format D2 F2) ⇒ Format (D1 ^ D2) (F1 · F2) where
    format (d1 ^ d2)  = format d1 ◇ format d2
```

In implementing the specification of section 3 we have simply replaced a type function by a functional type relation. Before we proceed let us take a look at an example translation.

```
    format (int ^ lit " is " ^ str)
=     {definition of format}
      show ◇ " is " ◇ id
=     {definition of '◇'}
    map_{Int→} (λs → map_Id (λt → map_{String→} (λu → s ++ t ++ u) id) " is ") show
=     {definition of map_{A→} and map_Id}
    (λs → (λt → (λu → s ++ t ++ u) · id) " is ") · show
=     {algebraic simplifications and β-conversion}
    λi → λu → show i ++ " is " ++ u
```

We obtain exactly the function one would have written by hand. Note that simplifications along these lines can always be performed at compile time since the first argument of *format* is essentially static (apart from *lit*'s string argument).

5 Back to continuation-passing style

It is instructive to compare our solution to the original one by Danvy (1998), which makes use of a *continuation* and an *accumulating argument*. Phrased as a Haskell type class, Danvy's solution reads:

```
class Format' D F | D → F where
    format' :: ∀A . D → (String → A) → (String → F A)
instance Format' LIT Id where
    format' (lit s) = λκ out → κ (out ++ s)
instance Format' INT (Int →) where
    format' int     = λκ out → λi → κ (out ++ show i)
```

https://doi.org/10.1017/S0956796802004367 Published online by Cambridge University Press

```
instance Format' STR (String →) where
    format' str        = λκ out → λs → κ (out ++ s)
instance (Format' D1 F1, Format' D2 F2) ⇒ Format' (D1 ^ D2) (F1 · F2) where
    format' (d1 ^ d2)  = λκ out → format' d1 (format' d2 κ) out
format                 :: (Format' D F) ⇒ D → F String
format d               = format' d id "".
```

Two remarks are in order. First, the instances do not require mapping functions, which explains why *Format'* is not declared a subclass of *Functor*, though morally the second argument of *Format'* is a functor. Secondly, *format'* $(d_1 \mathbin{\hat{}} d_2)$ can be simplified to *format'* $d_1 \cdot$ *format'* d_2, where "$\cdot$" is ordinary function composition. We will take these points up again in the following section.

So we interpret *format* d by values of type *F String* whereas Danvy employs values of type $\forall A.(A \to String) \to (A \to F\ String)$. An obvious question is, of course, whether the two approaches are equivalent. Here are functions that convert to and fro:

$$\begin{aligned} \alpha\ d &= \lambda\kappa\ out \to map\ (\lambda s \to \kappa\ (out +\!\!+ s))\ d \\ \gamma\ d' &= d'\ id\ \texttt{""}. \end{aligned}$$

The coercion function α introduces a continuation and an accumulating string, while γ supplies an initial continuation and an empty accumulating string.

It is easy to see that $\gamma \cdot \alpha = id$:

$$\begin{aligned} \gamma\ (\alpha\ d) &= \quad \{\text{definition of } \gamma \text{ and } \alpha\} \\ &\quad (\lambda\kappa\ out \to map\ (\lambda s \to \kappa\ (out +\!\!+ s))\ d)\ id\ \texttt{""} \\ &= \quad \{\beta\text{-conversion}\} \\ &\quad map\ (\lambda s \to id\ (\texttt{""} +\!\!+ s))\ d \\ &= \quad \{\text{algebraic simplifications}\} \\ &\quad map\ id\ d \\ &= \quad \{\text{functor laws}\} \\ &\quad d. \end{aligned}$$

When we try to prove the converse, $\alpha \cdot \gamma = id$,

$$\begin{aligned} \alpha\ (\gamma\ d') &= \quad \{\text{definition of } \alpha \text{ and } \gamma\} \\ &\quad \lambda\kappa\ out \to map\ (\lambda s \to \kappa\ (out +\!\!+ s))\ (d'\ id\ \texttt{""}), \end{aligned}$$

we are immediately stuck. There is no obvious way to simplify the final expression. Note, however, that d' has a polymorphic type, so we can appeal to the *parametricity theorem* (Wadler, 1989). The "free theorem" for $d' :: \forall A.(String \to A) \to (String \to F\ A)$ is that for all $\phi :: A_1 \to A_2$ and for all $\epsilon :: String \to A_1$,

$$map\ \phi \cdot d'\ \epsilon = d'\ (\phi \cdot \epsilon). \tag{1}$$

Loosely speaking, this rule allows us to shift a part of the continuation to the left.

https://doi.org/10.1017/S0956796802004367 Published online by Cambridge University Press

Continuing the proof we obtain:

$$\begin{aligned} &= \quad \{\text{parametricity (1): } \phi = \lambda s \to \kappa\ (out +\!\!+ s) \text{ and } \epsilon = id\} \\ &\quad \lambda\kappa\ out \to d'\ (\lambda s \to \kappa\ (out +\!\!+ s))\ \texttt{""}. \end{aligned}$$

We are stuck again. This time we require a rule that allows us to shift a part of the continuation to the right. Let us assume for the moment that for all $\epsilon :: String \to A$ and for all $\sigma :: String \to String$,

$$d'\ (\epsilon \cdot \sigma) = d'\ \epsilon \cdot \sigma. \tag{2}$$

Given this property, we can finish the proof:

$$\begin{aligned} &= \quad \{\text{proof obligation (2): } \epsilon = \kappa \text{ and } \sigma = \lambda s \to out +\!\!+ s\} \\ &\quad \lambda\kappa\ out \to d'\ (\lambda s \to \kappa\ s)\ (out +\!\!+ \texttt{""}) \\ &= \quad \{\text{algebraic simplifications and } \eta\text{-conversion}\} \\ &\quad d'. \end{aligned}$$

It remains to establish the proof obligation. Perhaps unsurprisingly, it turns out that the rule does not hold in general. The problem is that the accumulating argument has a too concrete type: it is a string, which we can manipulate at will. In the following instance, for example, the accumulator is replaced by an empty string.

```
data CANCEL       = cancel
instance Format' CANCEL Id where
    format' cancel = λκ out → κ ""
```

The effect of *cancel* is to discard the string produced by the directives to its left.

```
Main⟩ format' (int ^ lit " is " ^ cancel ^ str) 5 "five"
"five"
```

One might argue that the ability to define such a directive is an unwanted consequence of switching to continuation passing style. In that sense, rule (2) is really a proof obligation for the programmer. As a closing remark, note that we can achieve a similar effect in our setting using a "forgetful" variant of "$\diamond$":

$$\begin{aligned} f \triangleright g &= map\ (\lambda s \to map\ (\lambda t \to t)\ g)\ f \\ f \triangleleft g &= map\ (\lambda s \to map\ (\lambda t \to s)\ g)\ f. \end{aligned}$$

6 Applying a functor

Let us finally turn the code of section 4 into an executable Haskell program (for consistency, we stick to the lexical conventions of section 2). Recall that the instance declarations involving the type synonyms *Id* and "$\cdot$" are not legal since type synonyms must not be partially applied. Therefore, we are forced to introduce the two types via **newtype** declarations:

```
newtype Id A       = ide A
newtype (F · G) A  = com (F (G A)).
```

Alas, now *Id* and "$\cdot$" are new distinct types. In particular, the identities $Id\ A = A$ and $(F \cdot G)\ A = F\ (G\ A)$ do not hold any more: the type of *format* (*int* ^ *lit* `" is "` ^ *str*)

https://doi.org/10.1017/S0956796802004367 Published online by Cambridge University Press

is $((Int \rightarrow) \cdot Id \cdot (String \rightarrow))\ String$ rather than $Int \rightarrow String \rightarrow String$. In order to obtain the desired type we have to apply the functor $(Int \rightarrow) \cdot Id \cdot (String \rightarrow)$ to the type *String*. This type transformation is implemented by the following three parameter type class:

```
class (Functor F) ⇒ Apply F A B | F A → B where
    apply          :: F A → B
instance Apply (A →) B (A → B) where
    apply          = id
instance Apply Id A A where
    apply (ide a)  = a
instance (Apply G A B, Apply F B C) ⇒ Apply (F · G) A C where
    apply (com x)  = apply (map apply x).
```

The intention is that the type relation $Apply\ F\ A\ B$ holds iff $F\ A = B$. Consequently, B is uniquely determined by F and A, which is expressed by the functional dependency $F\ A \rightarrow B$ (again, do not confuse the dependency with a functional type). The class method *apply* always equals the identity function since a **newtype** has the same representation as the underlying type. Now, renaming the class method of *Format* to *formatx* we arrive at the true definition of *format*:

```
format   :: (Format D F, Apply F String A) ⇒ D → A
format d = apply (formatx d).
```

7 Haskell can do it (almost) without type classes

Given the title of the pearl this final twist is perhaps unexpected. We can quite easily eliminate the *Format* class by *specializing format* to the various types of directives: for each $d :: D$ we introduce a new directive $\underline{d} :: Format_D\ String$ given by $\underline{d} = formatx\ d$ – we omit the underlining in the sequel and just reuse the original names.

```
lit      :: String → Id String
lit s    = ide s

int      :: (Int →) String
int      = λi → show i

str      :: (String →) String
str      = λs → s

format   :: (Apply F String A) ⇒ F String → A
format d = apply d

formatIO :: (Apply F (IO ()) A) ⇒ F String → A
formatIO d = apply (map putStrLn d)
```

So *int* is just *show* (albeit with a less general type), *str* is just *id*, and *format* is just *apply* (again with a less general type). Furthermore, instead of '^' we use '⋄'. We have also defined a variant of *format* that outputs the string to the standard output device. This function nicely demonstrates how to define one's own variable-argument

https://doi.org/10.1017/S0956796802004367 Published online by Cambridge University Press

functions on top of *format*. Here is an example session that illustrates the use of the new unparsing combinators:

```
Main⟩ :type (int ◇ lit " is " ◇ str)
((Int →) · Id · (String →)) String
Main⟩ :type format (int ◇ lit " is " ◇ str)
Int → String → String
Main⟩ format (int ◇ lit " is " ◇ str) 5 "five"
"5 is five"
Main⟩ format (show ◇ lit " is " ◇ show) 5 "five"
"5 is \"five\""
Main⟩ format (lit "sum " ◇ show ◇ lit " = " ◇ show) [1..10] (sum [1..10])
"sum [1,2,3,4,5,6,7,8,9,10] = 55".
```

Note the use of *show* in the last two examples. In fact, we can now seamlessly integrate Haskell's predefined unparsing function with our own routines. As an illustration, consider the following directive for unparsing a list of values:

```
list                :: (A →) String → ([A] →) String
list d []           =  "[]"
list d (a : as)     =  "[" ++ d a ++ rest as
  where rest []     =  "]"
        rest (a : as) = ", " ++ d a ++ rest as.
```

To format a string, for instance, we can now either use the directive *str* (emit the string literally), *show* (put the string in quotes), or *list show* (show the string as a list of characters). Likewise, for formatting a list of strings we can choose between *show*, *list str*, *list show*, or *list* (*list show*).

Can we also get rid of *Id*, "·" and consequently of the class *Apply*? Unfortunately, the answer is in the negative. Though all directives possess legal Haskell 98 types, Haskell's kinded first-order unification gets in the way when we combine the directives. Loosely speaking, the **newtype** constructors are required to direct the type checker. Interestingly, Danvy's solution seems to require a less sophisticated type system: the combinators possess ordinary Hindley-Milner types. However, this comes at the expense of type safety as a closer inspection reveals. The critical combinator is the one for concatenating directives, which possesses the following *rank − 2 type* (consider the instance declaration for "^" in section 5):

$$\begin{aligned}(\cdot) \;::\; \forall F\ G\ .(\forall X\ .(String \to X) \to (String \to F\ X)) \\ \to (\forall Y\ .(String \to Y) \to (String \to G\ Y)) \\ \to (\forall Z\ .(String \to Z) \to (String \to F\ (G\ Z))).\end{aligned}$$

Since "·" amounts to function composition we can generalize (or rather, weaken) the type to

$$\begin{aligned}(\cdot) \;::\; \forall A\ B\ C\ .(B \to C) \\ \to (A \to B) \\ \to (A \to C).\end{aligned}$$

Since Danvy's combinators furthermore do not employ mapping functions, they can be made to run in a language with a Hindley-Milner type system. Or course,

https://doi.org/10.1017/S0956796802004367 Published online by Cambridge University Press

weakening the types has the immediate drawback that, for instance, the non-sensible call *format* (*const* · *length* · *run*) where *run* $k = k$ "" is well-typed.

8 Summary and further reading

Do you have a similar problem that involves capturing dependent types in Haskell? Here is a brief summary of the overall construction. Let us assume that we are given a dependently typed function $f :: (t :: T) \to F\ t$ where $F :: T \to \star$ is the dependent type. The central step is to lift the elements of T to the type level such that for every element t of T there is a corresponding type $\bar{t}$. Through the lifting we obtain a family of functions $f_{\bar{t}} :: \bar{t} \to F_{\bar{t}}$ and a family of types $F_{\bar{t}} :: \star$ both indexed by type. These families can be immediately represented in Haskell using a two argument type class with a functional dependency. If the function f is defined by structural recursion, then each equation gives rise to an instance declaration (the corresponding equation of the dependent type F goes into the instance head albeit in a relational form).

Do you want to delve deeper into the world of singleton types, multiple-parameter type classes, and functional dependencies? In this case, I recommend reading Hallgren (2001), Neubauer *et al.* (2001), Gasbichler *et al.* (2002) and McBride (2002).

Acknowledgements

Thanks are due to Dave Clarke, Johan Jeuring, Andres Löh, Doaitse Swierstra, Peter Thiemann and Stephanie Weirich for their comments on a previous draft of this paper.

References

Augustsson, L. (1999) Cayenne – a language with dependent types. *SIGPLAN Notices*, **34**(1): 239–250.

Danvy, O. (1998) Functional unparsing. *J. Functional Programming*, **8**(6): 621–625.

Gasbichler, M., Neubauer, M., Sperber, M. and Thiemann, P. (2002) Functional logic overloading. *Proceedings 29th Annual ACM SIGPLAN-SIGACT Symposium on Principles of Programming Languages (POPL' 02)*, Portland, OR.

Hallgren, T. (2001) Fun with functional dependencies. *Proceedings Joint CS/CE Winter Meeting*, pp. 135–145. Department of Computing Science, Chalmers, Göteborg, Sweden.

Jones, M. P. (2000) Type classes with functional dependencies. In: Smolka, G., (editor), *Proceedings 9th European Symposium on Programming, ESOP 2000: Lecture Notes in Computer Science 1782*, pp. 230–244. Berlin, Germany. Springer-Verlag.

McBride, C. (2002) Faking it (simulating dependent types in Haskell). *J. Functional Programming*. To appear.

Neubauer, M., Thiemann, P., Gasbichler, M. and Sperber, M. (2001) A functional notation for functional dependencies. In: Hinze, R. (editor), *Proceedings 2001 ACM SIGPLAN Haskell Workshop*, pp. 101–120. Firenze, Italy.

Wadler, P. (1989) Theorems for free! *Fourth International Conference on Functional Programming Languages and Computer Architecture (FPCA'89)*, pp. 347–359. London, UK, Addison-Wesley.